
Abstract. We compute obstructions to the Hasse principle for curves, using
the example of elliptic curves. The computations are done using MAGMA.

1. The Algorithm

The method we will use to compute examples of obstructions to the Hasse Principle
for curves is described in [1]. We will be working with coverings of elliptic curves
which are genus 1 curves of the form

X : y2 = f(x), with f(x) ∈ Z[x].

As a first example, we will work over Qp (later we will want to work over the
localization of a number field).

If we denote Xp := X × Qp then the first step of the method is to compute an Fp

basis of Pic(Xp)/nPic(Xp) for a certain n.

This is accomplished by finding such a basis on the elliptic curve E which is the
Jacobian of X. Thus in particular, we compute explictly a parameterization

ρ : E(Qp) → X(Qp).

Firstly we set up the curve X in MAGMA. We define it intially over the rationals
and later find points on it over Qp.

p:=5;n:=3;
Qp:=pAdicField(p);
P<x,y,z>:=ProjectiveSpace(Rationals(),2);
PQp<y1>:=PolynomialRing(Qp);
X:=Curve(P,x^3*y^2+x^3*z^2-z^5);

In order to construct the parameterization ρ above we need to find a Qp rational
point pt on Xp. We do this by solving explicitly the equation defining X, over Qp.

i:=1;
repeat
ex,root:=HasRoot((Qp!i)^3*y1^2+(Qp!i)^3-1);
i:=i+1;
until ex eq true;
pt:=[Qp!i,root,Qp!1];

Now we are able to compute the elliptic curve E and a map toE (the inverse of ρ)
which takes the point pt to the origin of E. We also ensure that E is defined over
Qp.

E,toE:=EllipticCurve(X,pt);
EQp:=ChangeRing(E,Qp);

Next we compute the order of an Fp basis for E(Qp)/nE(Qp). This is given by the
formula

dimFpE(Qp)/nE(Qp) = dimFpE(Qp)[n] + δ,

where E(Qp)[n] is the Qp rational n-torsion of E and where δ is 0 for p 6= n and 1
if p = n.

G:=TorsionSubgroupScheme(E,n);
BasSize:=0;
while p^BasSize ne #G do
BasSize:=BasSize+1;
end while;
if n eq p then BasSize:=BasSize+1; end if;

The final step of the algorithm finds “random” Qp-rational points on Xp, maps
them to E via toE and checks whether they already lie in the Fp span of the points
found so far. If not, then they are added to the basis until it has the required
number of elements, as computed above.

To check whether a newly found point PonE is in the Fp span of the points in the
basis so far {P1, P2, . . . , Pr}, we construct the vector space V of dimension r over
Fp and iterating through all the vectors x = (x1, x2, . . . , xr) ∈ V we check that
PonE is not in the same class of E(Qp)/nE(Qp) as x1P1 + x2P2 + · · ·+ xrPr.

EqToE:=DefiningEquations(toE);
Bas:=[];VBas:=[pt];
while #Bas ne BasSize do
ex,root:=HasRoot((Qp!i)^3*y1^2+(Qp!i)^3-1);
if ex eq true then
point:=[Qp!i,root,Qp!1];
PonE:=[Evaluate(EqToE[1],point),Evaluate(EqToE[2],point),Evaluate(EqToE[3],point)];
bool,PonE:=IsPoint(EQp,PonE);
different:=true;
GFp:=GaloisField(p);
V:=VectorSpace(GFp,#Bas);
for x in V do
LinComb:=Id(EQp);
for j:=1 to #Bas do
LinComb=LinComb+ElementToSequence(x)[j]*Bas[j];
end for;
P2:= PonE-LinComb;
if IsDivisibleBy(P2,n) eq true then different:=false; end if;
end for;
if different eq true then Bas[#Bas+1]:=PonE; VBas[#Bas+1]:=point; end if;
end if;
i:=i+1;
end while;

At the end of this algorithm, VBas contains the point pt as its first element,
followed by a complete Fp basis for Pic(Xp)/nPic(Xp).

